
Проектиране, симулиране и изследване на аритметични устройства върху свръхголеми програмируеми(FPGA) - чипове

В това упражнение се проектира и имплементира върху  FPGA - чип цифрово устройство с конвейерна структура, извършващо аритметичните операции Y(i) = A(i) * B(i) + C(i), където операндите и резултатът са двоични числа, едноименни членове на едномерни масиви. Структурната схема на устройството е показана по-долу: 

Такова аритметично устройство се използва изключително много като част от архитектурата на високопроизводителни цифрови устойства и системи в съвременните приложни области  - цифрова обработка на сигнали, обработка на изображения, матрични научни пресмятания – всички те, извършващи се в реално време. По-долу са показани схемите на цифров филтър  и устройство за  изчисления за задача за  Преобразуване на Фурие, използващи в структурите си (и то многократно) проектираното тук устройство.

I.Проектиране на аритметично конвейерно устройство върху FPGA - чип
Извършете проектирането и имплементирането на управляващото устройство с помощта на развойната среда в следната последователност:

1. Създайте с помощта на Project Navigator на  Xilinx ISE 9.21  проект с име Pipeline09 в работната папка на вашата учебна група.

За тази цел:

1.1. Стартирайте програмната система за проектиране Project Navigator чрез:

- иконата Xilinx ISE (Integrated Software Environment) от Desktop

- или :  Start => Programs => Xilinx ISE 9.2i =>  Project Navigator
1.2. Създайте нов работен проект чрез командата  на основното меню на Project Navigator:

File => New Project
1.3. В прозореца New Project Wizard  - Create New project посочете необходимите елементи на създавания проект:

- в полето Project Location посочете папка за проекта: C:\SPRL\gr62a  (за група 62а);
- запишете името на проекта – Pipeline09 в полето Project Name – в папката gr62a ще се създаде папка Pipeline09 за файловете на проекта;
- изберете начин на описание на проектираното устройство, който ще се прилага в главния source(входен)-модул на проекта – в полето Top-Level Module Type изберете HDL (Hardware Description Language – декларира се използване на език за описание на хардуер);

След натискане на бутона Next преминете към въвеждане на останалите необходими елементи на проекта в следващия прозорец New Project Wizard  - Device Properties:
- изберете фамилия от FPGA – чипове, която ще се използва от проекта - Device Family: Spartan3E; 
- изберете FPGA – чип за  реализиране на проектираното устройство - Device: XC3S100E;

- изберете вид корпус на чипа - Package: TQ144;
- изберете скорост(бързодействие) на използвания чип -  Speed Grade: -5;
- средството за проектиране на схемата на устройството от езиковото му описание е технологията на синтезиране на цифрова схема XST (VHDL/Verilog) - Xilinx Synthesis Technology , използваща като входен език за описание на устройството езикът от високо ниво VHDL (който  ние ще използваме) или Verilog:
- изберете програма за симулиране на проектираните устройства: ISE Simulator (VHDL/Verilog);
[image: image6.png]=

=|New Project Wizard - Device Properties T

Selectthe Device and Design Flow forthe Project

Propetty Name Value
Product Category Al v
Family SpartandE, v
Device. XC35100E v
Package o144 v
Speed 5 v
Top-Level Source Type: HDL

Syrthesis Tool %ST VHDLVerlog) v
Sitmulator ISE Simulator [VHDL Verlog) v
Preferred Language VHDL v
Enable Enhanced Design Summary

Enable Message Fiering (]

Display Incremertal Messages o





1.4. Пропуснете чрез натискане на бутона  Next двата следващи прозореца на New Project Wizard, които се използват за създаване и добавяне на source-файлове към проекта; ние ще използваме друг начин за включване в проекта на такива файлове.

1.5. В изобразения на екрана прозорец New Project Wizard –  Project Summary се извеждат за справка елементите и параметрите на проекта, който  се създава. При несъгласие с някои от тези параметри е възможно връщане към предните стъпки от процедурата на създаване на проекта чрез бутона Back. Бутонът Finish, който трябва да натиснете, на практика създава в избраната папка нужните начални файлове на проекта.

2. Създайте главен VHDL source - модул на проекта с име pipeline, съдържащ описанието на структурата на проектираното устройство на език VHDL. 
2.1 За тази цел изпълнете от главното меню на  Project navigator: 
Project => New Source…
2.2 В отворилия се прозорец  “New Source Wizard – Select Source Type” изберете тип на модула:  VHDL Module (файл с VHDL – текст) , а във полето “File name” въведете името на файла , който ще съдържа този модул : Pipeline:  

Преминете нататък с натискане на бутона Next.
2.3 В следващия се отворил прозорец – “New Source Wizard – Define Module” въведете името на архитектурното описание на този модул -   Structural . 

Преминете нататък с Next.
2.4 В прозореца New Source Wizard – Summary се извеждат за справка параметрите на създавания модул:

 Ако сте съгласни с горните параметри, натиснете Finish, с което на практика се създава в папката на проекта файлът pipeline.vhd (vhd е разширение в името, означаващо модул(файл) с текст на езика VHDL).

-В прозореца за редактиране на текст развойната среда извежда текстова рамка на създадения VHDL – модул.(Pipeline.vhd), която би могла да се допълни, за да формира пълноценно VHDL – описание на устройството. . (Ако прозорецът с текстовата рамка не е показан на екрана, кликнете двойно с мишката върху името на модула Pipeline.vhd в прозореца Sources на Project Navigator):

2.5 Вместо да  допълваме текстовата рамка, ще създадем наново пълния текст, описващ  проектираното устройство. За тази цел:

- Изтрийте текстa в прозореца за редактиране на развойната среда и запишете на негово място като ново съдържание на модула pipeline.vhd следващия по-долу  VHDL – текст:
library IEEE;

use IEEE.STD_LOGIC_1164.ALL;

use IEEE.STD_LOGIC_ARITH.ALL;

use IEEE.STD_LOGIC_UNSIGNED.ALL;

Library UNISIM;

use UNISIM.Vcomponents.all;

entity pipeline is

    port (A : in std_logic_vector (15 downto 0);



 B : in std_logic_vector (15 downto 0);



 C : in std_logic_vector (15 downto 0);



 Y : out std_logic_vector (35 downto 0);




 CLK : in std_logic;



 RESET : in std_logic);

end pipeline;

architecture Structural of pipeline is
component reg16

    port 
(DIN: in std_logic_vector (15 downto 0);

DOUT: out std_logic_vector (15 downto 0);

CLK: in std_logic;


RESET: in std_logic);

end component;

component reg36

    port (
DIN: in std_logic_vector (35 downto 0);

 DOUT: out std_logic_vector (35 downto 0);

 CLK: in std_logic;


 RESET: in std_logic);

end component;

component adder

    port (
OPERAND1, OPERAND2 : in std_logic_vector (35 downto 0);

 RESULT : out std_logic_vector (35 downto 0));

end component;


signal OUT_A, OUT_B, OUT_C, OUT_C_1 : std_logic_vector (15 downto 0);


signal OUT_A_1, OUT_B_1 : std_logic_vector (17 downto 0);


signal OUT_M, OUT_ADD, OUT_M_2, OUT_C_2 :

 std_logic_vector (35 downto 0);

begin


reg_A: reg16 port map (DIN => A, DOUT => OUT_A, CLK => CLK,  RESET => RESET);


reg_B: reg16 port map (DIN => B, DOUT => OUT_B, CLK => CLK, RESET => RESET);


reg_C: reg16 port map (DIN => C, DOUT => OUT_C, CLK => CLK, RESET => RESET);


multy: MULT18X18 port map (A => OUT_A_1, B => OUT_B_1, P => OUT_M);


reg_MUL: reg36 port map (DIN => OUT_M, DOUT => OUT_M_2, CLK => CLK, RESET => RESET);


reg_C_2: reg16 port map (DIN => OUT_C, DOUT => OUT_C_1, CLK => CLK, RESET => RESET);


add: adder port map (OPERAND1 => OUT_M_2, OPERAND2 => OUT_C_2, RESULT => OUT_ADD);


reg_out: reg36 port map (DIN => OUT_ADD, DOUT => Y, CLK => CLK, RESET => RESET);


OUT_A_1 <= "00" & OUT_A;


OUT_B_1 <= "00" & OUT_B;


OUT_C_2 <= x"00000" & OUT_C_1;
end Structural;

2.6 Запишете новия текст на модула във файла pipeline.vhd (чрез File => Save… или използвайки “дискетката” от лентата с инструменти) : 

3. Разгледайте текста на модула. В архитектурата му са декларирани три компонента в устройството pipeline : reg16, reg36, adder, представящи типовете елементи (блокове) на структурата. Четвъртият компонент (умножител с 18- разрядни операнди MULT18X18) не е деклариран, въпреки че е използван в структурата на устройството. Той, обаче, под същото име е дефиниран в библиотеката Library UNISIM на Project Navigator.

В прозореца Sources in Project под главния модул след разширяване се появяват подсказвания (с въпросителните знаци) за необходимо включване в проекта на описания на трите изброени по-горе компонента на структурата reg16, reg36, adder в самостоятелни модули, подчинени на главния (pipeline.vhd):

4. Създайте и добавете в проекта като  нови source – модули, VHDL-модулите adder.vhd, reg16.vhd, reg36.vhd, описващи компонентите на структурата като самостоятелни устройства със същите имена и входно/изходни сигнали, както в съответните декларации на компонентите в главния модул.  ( Project => New Source… и след това аналогично на стъпките за създаване на главния модул по-горе – т.2. В стъпка 2.3 не се променя полето Architecture Name – остава за всички модули Behavioral)
Новите, прибавени модули на проекта трябва да имат следните съдържания (текстове):
· На модула adder.vhd, описващ суматора от структурата :
library IEEE;

use IEEE.STD_LOGIC_1164.ALL;

use IEEE.STD_LOGIC_UNSIGNED.ALL;

entity adder is

   port (
OPERAND1, OPERAND2 : in std_logic_vector (35 downto 0);

RESULT : out std_logic_vector (35 downto 0));

end adder;

architecture Behavioral of adder is

begin

RESULT <= OPERAND1 + OPERAND2;

end Behavioral;

-На модула reg16.vhd, описващ 16- разряден регистър от структурата :
library IEEE;

use IEEE.STD_LOGIC_1164.ALL;

entity reg16 is

    port (CLK: in std_logic;


    RESET: in std_logic;


    DIN: in std_logic_vector(15 downto 0);


    DOUT: out std_logic_vector(15 downto 0) );

end reg16;

architecture Behavioral of reg16 is

begin

process (CLK, RESET)

begin

   if RESET='1' then  

   

DOUT <= x"0000";

   elsif (CLK'event and CLK='1') then  

  

DOUT <= DIN;

   end if;

end process;

end Behavioral;

-На модула reg36.vhd, описващ 36-разряден регистър от структурата:

library IEEE;

use IEEE.STD_LOGIC_1164.ALL;

entity reg36 is

   

 port (CLK: in std_logic;



 RESET: in std_logic;



 DIN: in std_logic_vector(35 downto 0);



 DOUT: out std_logic_vector(35 downto 0) );

end reg36;

architecture Behavioral of reg36 is

begin

process (CLK, RESET)

begin

   
if RESET='1' then  

   


DOUT <= (others => '0');

  
 elsif (CLK'event and CLK='1') then  

DOUT <= DIN;

  
 end if;

end process;

end Behavioral;

След записване на текстовете в модулите, в прозореца Sources in Project на местата на въпросителните би трябвало да се появят  имената на компонентите и имената на архитектурните им  описания - в скоби преди  имената на модулите:

5. Изпълнете процесите на проектиране на устройството. За тази цел:

5.1. В прозореца Processes (при избран  главен модул  pipeline в прозореца Sources и в режим за този прозорец Synthezis/Implementation) двойно  кликнете върху последния процес: Generate Programmilng File:

Project Navigator ще стартира всички процеси на проектиране в необходимата последователност. Индикатор за успешно завършило проектиране на устройството върху FPGA - чипа е поставената от развойната среда зелена отметка пред последната група процеси в прозореца Processes .
Какво извършват процесите на проектиране на развойната среда (от прозореца Processes for Current Source):

5.2. Изобразете и разгледайте на екрана на компютъра синтезираната от развойната среда  цифрова схема на устройството чрез:

- двойно кликване на Synthesize -> View RTL Schematics в прозореца Processes;

- влизане в по-подробните нива на схемата чрез командата с икона “стрелка надолу” от лентата с инструментите на Project Navigator:
[image: image16.png]Source Process Window  Help.
SREX Do £H
KD AAF BB





- Общ вид на най-високо ниво на схемата:

- Вид на схемата с основните й блокове (след влизане в по-подробно ниво):

- След като разгледате и проверите схемата, затворете прозореца на чертежа й (затворете модул pipeline.ngr на проекта).
5.4. Проверете разположението на използваните блокове от FPGA-чипа за проектираното устройство чрез програмата на развойната среда Floor Planner – ще я стартирате чрез двойно кликване на Implement Design->Place&Route->View/Edit Placed Design  в прозореца Processes (при избран  модул pipeline в прозореца Sources ).
На въпроса на развойната среда дали да създаде в този момент файл с данни за ограничения в работата й, зададени от потребителя (така нареченият UCF – файл), отговорете решително с No! (Такъв файл ще създадем и напълним с данни по-късно по време на това упражнение):

На схемата, извеждана от програмата Floorplanner след изпълняване на zoom + операции (чрез команда с икона [image: image20.png]


 от лентата с инструментите), например се виждат (отбелязани в зелен  цвят) използваният за устройството, съществуващ в структурата на FPGA - чипа 18- разряден умножител и синтезираният от конфигурируеми логически блокове 36 – разряден суматор:

· Затворете прозореца със схемата на програма Xilinx Floorpranner.
II. Симулиране на работата на проектираното устройство:

Проектираното устройство би трябвало да извършва изчисленията върху масиви от данни А, B, C в конвейерен режим – да започва изчислението на всяка поредна итерация Y(i)=A(i)*B(i)+C(i)  преди да е завършило изчислението на предната итерация, с цел намаляване на общото време Tз за извършване на изчислението за  всички членове на масивите (за цялата задача):

За да извършите симулиране на проектираното устройство, преминете в режим на работа на развойната среда Post-Route Simulation в прозореца Sources чрез избор от падащото меню ( това означава, че ще се извърши симулиране на устройството, реализирано върху FPGA – чипа, със схемотехническите параметри на настройваемите блокове на чипа):

6. Включете в проекта нов модул (чрез  командите Project => New Source…) от тип Test Bench Waveform, с име на модула testpipe, който ще служи за съхраняване на стимулите (стойности във времето на входните сигнали) за бъдещата симулация на проектираното устройство. 
6.1. Укажете чрез прозореца New Source Wizard – Associate Source, че този модул трябва да бъде подчинен (асоцииран) с главния VHDL-модул , описващ архитектурата на устройството – а именно  pipeline.
6.2. Задайте в появилия се прозорец Initialize Timing and Clock Wizard  параметрите на симулацията:

- период на тактовия сигнал 20ns чрез продължителности на тактовия импулс и на паузата между импулсите: Clock High Time = 10 ns, Clock Low Time = 10 ns  ( колкото е периодът на синхроимпулсите, сенерирани от генератора на развойната платка);

- моменти на промяна и отчитане на стойностите на сигналите спрямо нарастващия фронт на тактовия сигнал – Input Setup Time = 3 ns, Output Valid Delay = 2 ns;

- активен Rising Edge (в регистрите на устройството ще се записват нови данни в моментите на нарастващите фронтове на тактовия сигнал).
- продължителност на интервала от време за симулация (чрез Initial Length of Test Bench) - 1000 ns
6.3. Задайте активен Single Clock чрез съответния радиобутон (ако не е вече активиран) и изберете чрез падащото меню за тактов сигнал при симулацията сигнала CLK (ако не е вече избран):

- Затворете чрез бутона Finish прозореца с параметрите на симулацията. В прозореца за редактиране на Project Navigator ще се появят времедиаграми на сигналите на симулираното устройство, които ще се съхраняват в модула testpipe.tbw
6.4. Задайте за по-голямо удобство в показалия се прозорец с времедиаграми на модула testpipe стойностите на сигналите да се показват като десетични числа – чрез кликване върху времедиаграма с десен бутон и съответен избор Decimal Unsigned от отворилото се контексно меню:

[image: image25.png]ZoomIn
Zoom Ot
Zoom Full View

addMeasure
addMarker

Decimal (Signed)

v Decimal (Unsigned)
Hexadecimal

Binary
ascn

arid
Set End of Test Bench,





6.5. Задайте няколко набора от стойности на входните сигнали (А, В, С) в съседни тактови периоди и подходящи стойности на управляващия сигнал RESET, за да функционира устройството. 
- За еднобитовите входни сигнали (в нашия случай  - RESET) кликването в реда на времедиаграмата на сигнала в някой период на тактовия сигнал задава алтернативна промяна на стойността му от началото на този период. (За да стане това, курсорът не трябва да е върху тъмносивия интервал); В началото на интервала от време за симулация RESET трябва да бъде установен в ‘1’, за да се нулират съдържанията на регистрите за данни на симулираното устройство, а след това трябва да бъде установен в ‘0’, за да могат да бъдат записвани други данни в регистрите.

- Нова стойност на входен сигнал, който не е едноразряден, се задава крез кликване с ляв бутон в съответния периoд на тактовия сигнал на времедиаграмата и въвеждане на стойността му чрез прозореца Set Value:

След въвеждане на стойностите на входните сигнали, прозорецът за редактиране за модула от тип Test Bench Waveform би било добре да изглежда по принцип така (новите стойности на аргументите А, В, С да постъпват на съответните входове на устройството в едни и същи моменти):

6.6. Изпълнете команда File -> Save (или кликнете върху “дискетката” от лентата с инструменти, с което информацията за въведените от вас времедиаграми на входните сигнали се записва в създадения Test Bench Waveform – модул (в случая testpipe.tbw). След записа в прозореца  Sources би трябвало да се появи името на този модул:

7. Стартирайте симулатора на развойната среда Xilinx ISE Simulator. За тази цел:

7.1. Активирайте Test Bench Waveform –  модула testpipe  в прозореца Sources;

7.2 .Разширете Xilinx ISE Simulator и стартирайте чрез двойно кликване в прозореца  Processes  командата Simulate  Post-Place&Route Model:

Прозорецът с резултати (времедиаграми) от работата на симулатора след  zoom + (чрез команда с икона [image: image30.png]


 от лентата с инструментите) и преобразуване на данните в десетичен вид би трябвало да изглежда  така:

3x2 + 3 = 9;  4x3 + 2 = 14; 5x4 + 1 = 21
Времедиаграмите и стойностите на аргументите и резултата илюстрират типично конвейерно функциониране на аритметично устройство – преди да се получи на изхода стойността на резултат, се въвеждат на входовете стойностите на следващите аргументи и започват операциите с тях; тази организация е с цел  по-добро използване на функционалните блокове (в случая - умножител и суматор) и оттам по-голяма производителност на изчисленията. 
7.3. Ако нямате забележки към получените резултати от изчисленията, прекратете работата на симулатора, като затворите прозореца с времедиаграмите.
III. Проверка на работата на проектираното устройство върху развойната платка.

Ще допълним досега направения проект за развойната среда така, че да можем да проверим функционирането на проектираното устройство върху развойната платка.

Освен досега създаденото аритматично устройство, върху чипа ще имплементираме и устройство, което управлява индицирането на резултатите от аритметичните операции в десетичен вид върху седемсегментни индикаторни елементи на платката. Ще използваме ключовете на платката за ръчно въвеждане на стойности за аргументите A, B, C.
8. Преминете обратно в режим на работа на развойната среда: Synthesis/Implementation (установява се чрез падащото меню в прозореца Sources)
8.1. Добавете в проекта vhd – модул, описващ устройство, управляващо извеждането на информация (десетични числа) върху 7 – сегментните индикаторни елементи на развойната платка – чрез Project... => New Source... => в отворилия се прозорец въведете VHDL Module и за  File name:  SevenSegInd и име на архитектурата му: Behavioral; изтрийте текстовата рамка на модула в прозореца за редактиране на Project navigator и въведете следния нов текст за него:
library IEEE;

use IEEE.STD_LOGIC_1164.ALL;

use IEEE.STD_LOGIC_ARITH.ALL;

use IEEE.STD_LOGIC_UNSIGNED.ALL;

entity SevenSegInd is


port( Yindication : in std_logic_vector(7 downto 0);




CLK : in std_logic;




CA : out std_logic;




CB : out std_logic;




CC : out std_logic;




CD : out std_logic;      






CE : out std_logic;






CF : out std_logic;




CG : out std_logic;




AN0 : out std_logic;





AN1 : out std_logic;







AN2 : out std_logic;





AN3 : out std_logic);

end SevenSegInd;

architecture Behavioral of SevenSegInd is

signal BCD : std_logic_vector(11 downto 0);

signal inputHex: std_logic_vector (3 downto 0);

signal segment : std_logic_vector (6 downto 0);

---------------------------------------

signal time_counter : std_logic_vector(19 downto 0); 

---------------------------------------

signal CMUX_control : std_logic_vector(1 downto 0);

---------------------------------------

signal digit_3 : std_logic_vector(3 downto 0);

signal digit_2 : std_logic_vector(3 downto 0);

signal digit_1 : std_logic_vector(3 downto 0);

signal digit_0 : std_logic_vector(3 downto 0);

------------- function binary To BCD ------------------------------------------------------


function BinaryToBCD(binary : std_logic_vector(7 downto 0) ) return std_logic_vector is



variable i : integer := 0;



variable bcd_code : std_logic_vector(19 downto 0) := x"000" & binary;



variable bcd_finally : std_logic_vector(11 downto 0) := (others => '0');


begin 



for i in 0 to 7 loop




bcd_code(19 downto 1) := bcd_code(18 downto 0);




bcd_code(0) := '0';




if(i<7)then





if(bcd_code(11 downto 8) > "0100")then






bcd_code(11 downto 8) := bcd_code(11 downto 8) + "0011";





end if;





if(bcd_code(15 downto 12) > "0100")then






bcd_code(15 downto 12) := bcd_code(15 downto 12) + "0011";





end if;





if(bcd_code(19 downto 16) > "0100")then






bcd_code(19 downto 16) := bcd_code(19 downto 16) + "0011";





end if;




end if;



end loop;



bcd_finally := bcd_code(19 downto 8);


return bcd_finally; 


end BinaryToBCD;

---------------------------------------------------------------------------------

begin

process (CLK)

begin

if CLK = '1' and CLK'event then

     

time_counter <= time_counter +1;

end if;

end process;


CMUX_control <= time_counter(13 downto 12);

BCD <= BinaryToBCD(Yindication);

digit_0 <= BCD(3 downto 0);

digit_1 <= BCD(7 downto 4);

digit_2 <= BCD(11 downto 8);

digit_3 <= "0000";

process (CMUX_control, digit_3, digit_2, digit_1, digit_0)

begin

   case CMUX_control is

      when "00" => inputHex <= digit_0(3 downto 0);

      when "01" => inputHex <= digit_1(3 downto 0);

      when "10" => inputHex <= digit_2(3 downto 0);

      when "11" => inputHex <= digit_3(3 downto 0);

      when others => inputHex <= digit_0(3 downto 0);

   end case;

end process;

process (CMUX_control)

begin

   case CMUX_control is

      when "00" => AN0 <= '0'; AN1 <= '1';AN2 <= '1';AN3 <= '1';

      when "01" => AN0 <= '1'; AN1 <= '0';AN2 <= '1';AN3 <= '1';

      when "10" => AN0 <= '1'; AN1 <= '1';AN2 <= '0';AN3 <= '1';

      when "11" => AN0 <= '1'; AN1 <= '1';AN2 <= '1';AN3 <= '0';

      when others => AN0 <= '0'; AN1 <= '1';AN2 <= '1';AN3 <= '1';

   end case;

end process;

process (inputHex)

begin

case inputHex is







when "0001" => Segment <= "1001111";   --1






when "0010" => Segment <= "0010010";   --2






when "0011" => Segment <= "0000110";   --3






when "0100" => Segment <= "1001100";   --4






when "0101" => Segment <= "0100100";   --5






when "0110" => Segment <= "0100000";   --6






when "0111" => Segment <= "0001111";   --7






when "1000" => Segment <= "0000000";   --8






when "1001" => Segment <= "0000100";   --9






when others => Segment <= "0000001";   --0





end case;

end process;

 CA <=  segment(6);

 CB <=  segment(5);

 CC <=  segment(4);

 CD <=  segment(3);

 CE <=  segment(2);

 CF <=  segment(1);

 CG <=  segment(0);

end Behavioral;

Запишете във файла SevenSegInd.vhd новия текст на модула (File -> Save… или чрез “дискетката” от лентата с инструменти).

Блокова схема на устройството, описано с горния текст изглежда така(схемата е синтезирана от развойната среда):

Горният VHDL - текст описва цифрово устройство, което преобразува двоичния код в младшите 8 разряда на изхода на аритметичното устройство Y(35 downto 0)  в BCD (десетичен код, десетичните цифри на който са представени двоично) с три десетични цифри. 
 Това устройство е описано поведенчески – чрез функция на VHDL, чийто текст отново показваме:   
function BinaryToBCD(binary : std_logic_vector(7 downto 0) ) return std_logic_vector,  декларирана и дефинирана в декларативната част на VHDL – архитектурата:
------------- function binary To BCD ------------------------------------------------------


function BinaryToBCD(binary : std_logic_vector(7 downto 0) ) return std_logic_vector is



variable i : integer := 0;



variable bcd_code : std_logic_vector(19 downto 0) := x"000" & binary;



variable bcd_finally : std_logic_vector(11 downto 0) := (others => '0');


begin 



for i in 0 to 7 loop




bcd_code(19 downto 1) := bcd_code(18 downto 0);




bcd_code(0) := '0';




if(i<7)then





if(bcd_code(11 downto 8) > "0100")then






bcd_code(11 downto 8) := bcd_code(11 downto 8) + "0011";





end if;





if(bcd_code(15 downto 12) > "0100")then






bcd_code(15 downto 12) := bcd_code(15 downto 12) + "0011";





end if;





if(bcd_code(19 downto 16) > "0100")then






bcd_code(19 downto 16) := bcd_code(19 downto 16) + "0011";





end if;




end if;



end loop;



bcd_finally := bcd_code(19 downto 8);


return bcd_finally; 


end BinaryToBCD;

---------------------------------------------------------------------------------

Устройството изобразява десетичните цифри върху 7 – сегментните индикаторни елементи на развойната платка, като преобразува двоичните кодове на тези десетични цифри в кодове, необходими за управлението на съответните конфигурации от сегменти (светодиоди):

[image: image33.emf]
Отново показваме VHDL – текста, който описва кодиращата схема, извършваща това преобразуване:
process (inputHex)

begin

case inputHex is







when "0001" => Segment <= "1001111";   --1






when "0010" => Segment <= "0010010";   --2






when "0011" => Segment <= "0000110";   --3






when "0100" => Segment <= "1001100";   --4






when "0101" => Segment <= "0100100";   --5






when "0110" => Segment <= "0100000";   --6






when "0111" => Segment <= "0001111";   --7






when "1000" => Segment <= "0000000";   --8






when "1001" => Segment <= "0000100";   --9






when others => Segment <= "0000001";   --0





end case;

end process;

Четирите седемсегментни индикаторни елементи на развойната платка имат общи

 управляващи сигнали за катодите на сегментите: [image: image34.emf]
и индивидуални управляващи сигнали за анодите им: [image: image35.emf].
[image: image36.emf]
Поради общите сигнали за катодите, налага се устройството, управляващо индикацията, във всеки момент да подава сигналите само към един елемент и да мултиплексира (сменя) във времето тези сигнали за четирите елемента и то с такава честота, че да не се забелязва примигването им:
[image: image37.emf]
Устройството, управляващо индикацията всъщност се състои от два мултиплексера, които под управлението на общ сигнал CMUX_control сменят във времето изходните си сигнали: digit_i (от 0 до 3) – за катодите  и  ANi (от 0 до 3) – за анодите на индикаторните елементи. По-долу отново показваме VHDL – текста, описващ тези два мултиплексера:
begin

if CLK = '1' and CLK'event then

     

time_counter <= time_counter +1;

end if;

end process;


CMUX_control <= time_counter(13 downto 12);

BCD <= BinaryToBCD(Yindication);

digit_0 <= BCD(3 downto 0);

digit_1 <= BCD(7 downto 4);

digit_2 <= BCD(11 downto 8);

digit_3 <= "0000";

process (CMUX_control, digit_3, digit_2, digit_1, digit_0)

begin

   case CMUX_control is

      when "00" => inputHex <= digit_0(3 downto 0);

      when "01" => inputHex <= digit_1(3 downto 0);

      when "10" => inputHex <= digit_2(3 downto 0);

      when "11" => inputHex <= digit_3(3 downto 0);

      when others => inputHex <= digit_0(3 downto 0);

   end case;

end process;

process (CMUX_control)

begin

   case CMUX_control is

      when "00" => AN0 <= '0'; AN1 <= '1';AN2 <= '1';AN3 <= '1';

      when "01" => AN0 <= '1'; AN1 <= '0';AN2 <= '1';AN3 <= '1';

      when "10" => AN0 <= '1'; AN1 <= '1';AN2 <= '0';AN3 <= '1';

      when "11" => AN0 <= '1'; AN1 <= '1';AN2 <= '1';AN3 <= '0';

      when others => AN0 <= '0'; AN1 <= '1';AN2 <= '1';AN3 <= '1';

   end case;

end process;

8.2. Добавете в проекта нов vhd – модул, описващ устройство, намаляващо честотата на тактовия сигнал към аритметичното устройство (с цел по-добро проследяване на функционирането на аритметичното устройство).
- Изпълнете от основното меню на Project Navigator : Project => New source…, в отворилия се прозорец  New Source Wizard  укажете, че ще създавате VHDL module,  въведете за име на новия модул ClockDivider и след като пропуснете прозореца за предварително деклариране на входно-изходни сигнали, изтрийте текстовата рамка на модула в прозореца за редактиране на Project navigator и въведете следния нов текст за него:
library IEEE;

use IEEE.STD_LOGIC_1164.ALL;

use IEEE.STD_LOGIC_ARITH.ALL;

use IEEE.STD_LOGIC_UNSIGNED.ALL;
entity ClockDivider is

    Port (    CLK : in  STD_LOGIC;

             ClkDiv : out  STD_LOGIC );

end ClockDivider;

architecture Behavioral of clockDivider is


signal COUNT: STD_LOGIC_VECTOR (28 downto 0) ;

begin


process (CLK)


begin



if CLK'event and CLK = '1' then


 

COUNT <= COUNT +1;



end if;



end process;

     ClkDiv <= COUNT(0);    

end Behavioral;
Прегледайте горния текст – той описва двоичен брояч, който брои тактовите импулси от входния сигнал CLK. 
От горния VHDL – текст развойната среда (WebPACK) синтезира  схема на брояч, съставена  от последователно свързани Т – тригери:

Схемата функционира така: всеки от тригерите (както функционира Т – тригер) се обръща (от 0 към 1 и от 1 към 0) по всеки нарастващ фронт на тактовия сигнал на входа му Т (който е изходът COUNT(i)  на предния, по-младши тригер, с инверсия). 

Графично това е представено по-долу, чрез диаграмите на изходните сигнали COUNT(i) на първите (младшите) тригери на брояча, получени от симулацията чрез ISE Simulator на брояч, описан с горния VHDL -текст:

От времедиаграмите може да се види, че: всеки тригер от брояча генерира с изхода си COUNT(i) импулсна поредица с два пъти по-малка честота, от колкото предния, по-младши тригер COUNT(i-1). 
Според дадения по-горе VHDL – текст, броячът ще генерира импулси с тактова честота, два пъти по-малка от входната (за генератора на развойната платка, която използваме – 50/2 = 25 Mhz). По-нататък в упражнението ще намалим значително честотата на тези импулси, за да можем да изследваме по-лесно функционирането на аритметичното устройство и устройството за извеждане върху 7 – сегментните индикатори на   десетични числа.
- Запишете във файла ClockDivider.vhd текста на горното устройство (File -> Save…).

8.3. Добавете в проекта главен vhd – модул main.vhd,  с име на архитектурата му: Structural, описващ структурата на системата от устройства, която ще имплементираме върху развойната платка: аритметичното устройство pipeline, устройството за управление на индикацията SevenSegInd, устройството за намаляване на тактовата честота на генератора на импулси на платката ClockDivider . VHDL – текстът на този главен модул е даден след общата блокова схема на системата:

VHDL – текст на главния модул на проекта main:
library IEEE;

use IEEE.STD_LOGIC_1164.ALL;

use IEEE.STD_LOGIC_ARITH.ALL;

use IEEE.STD_LOGIC_UNSIGNED.ALL;

entity Main is

port (Reset : in std_logic;


   BoardClock   : in std_logic;



IndClock : out std_logic;



 Ain : in std_logic_vector (3 downto 0);



 Bin : in std_logic_vector (3 downto 0);



 Cin : in std_logic_vector (3 downto 0);

      CA : out std_logic;



CB : out std_logic;



CC : out std_logic;



CD : out std_logic;      





CE : out std_logic;





CF : out std_logic;



CG : out std_logic;



AN0 : out std_logic;




AN1 : out std_logic;






AN2 : out std_logic;




AN3 : out std_logic

);

end Main;

architecture Structural of Main is

component pipeline is

  port(A : in std_logic_vector (15 downto 0);



 B : in std_logic_vector (15 downto 0);



 C : in std_logic_vector (15 downto 0);



 Y : out std_logic_vector (35 downto 0);




 CLK : in std_logic;



 RESET : in std_logic

);

end component;

component SevenSegInd is


port( Yindication : in std_logic_vector(7 downto 0);




CLK : in std_logic;




CA : out std_logic;




CB : out std_logic;




CC : out std_logic;




CD : out std_logic;      






CE : out std_logic;






CF : out std_logic;




CG : out std_logic;




AN0 : out std_logic;





AN1 : out std_logic;







AN2 : out std_logic;





AN3 : out std_logic);

end component;


component ClockDivider is

    Port (    CLK : in  STD_LOGIC;

             ClkDiv : out  STD_LOGIC );


end component;

signal A, B, C : std_logic_vector (15 downto 0);

signal Y: std_logic_vector (35 downto 0);

signal InternalClk : std_logic;

signal Yindication: std_logic_vector (7 downto 0);

begin

pipe: pipeline port map ( A => A, B => B, C => C, Y => Y, CLK => InternalClk, Reset => Reset);

SevInd: SevenSegInd port map (Yindication => Yindication, CLK => BoardClock,CA => CA, 

CB => CB, CC => CC, CD => CD, CE => CE, CF => CF, CG => CG, AN0 => AN0, AN1 => AN1,

AN2 => AN2, AN3 => AN3);

ClkDiv: ClockDivider port map (Clk => BoardClock, ClkDiv => InternalClk);

A <= x"000" & Ain;

B <= x"000" & Bin;

C <= x"000" & Cin;

IndClock <= InternalClk;

Yindication <= Y(7 downto 0);

end Structural;

-------------------------------------------------------------------------------------------------------------------------------------

Древните гърци твърдят, че с помощта на числата може да се опише всичко. Според нас това далеч не е така!

9. Разположете входните и изходните сигнали на устройството върху определени изводи на използвания FPGA -чип от  развойната платка  с цел  да могат да се въвеждат входни данни от определени ключове и резултатите от аритметичните операции да се индицират чрез седемсегментните индикаторни елементи, които върху платката са свързани с определените крачета (пинове) на чипа.

9.1.За въвеждане на горната информация (за разполагане на входно/изходните сигнали по крачетата на чипа) добавете към проекта нов source - модул от тип Implementation constraints file (с име main_Ucf) чрез Project => New Source... Потвърдете в отворилия се следващ прозорец New Sources Wizard -  Associate Source, че този нов модул ще бъде асоцииран (подчинен) на главния VHDL - модул main. 

В прозореца Sources in Project ще получите следната актуална конфигурация за включените досега модули в проекта:

9.2. При избран source-модул main_Ucf в прозореца Sources двойно кликнете върху Edit Constraints (Text) в прозореца Processes:

9.3. В полето за редактиране на текст вдясно въведете следния текст, който задава разполагането на всеки от входно/изходните сигнали  (NET) по съответното краче на чипа (LOC):
NET "Ain<0>"  LOC = "p18"  ; 

NET "Ain<1>"  LOC = "p12"  ; 

NET "Ain<2>"  LOC = "p10"  ; 

NET "Ain<3>"  LOC = "p6"  ; 

NET "AN0"  LOC = "p34"  ; 

NET "AN1"  LOC = "p33"  ; 

NET "AN2"  LOC = "p32"  ; 

NET "AN3"  LOC = "p26"  ; 

NET "Bin<0>"  LOC = "p38"  ; 

NET "Bin<1>"  LOC = "p36"  ; 

NET "Bin<2>"  LOC = "p29"  ; 

NET "Bin<3>"  LOC = "p24"  ; 

NET "BoardClock"  LOC = "p54"  ; 

NET "CA"  LOC = "p25"  ; 

NET "CB"  LOC = "p16"  ; 

NET "CC"  LOC = "p23"  ; 

NET "CD"  LOC = "p21"  ; 

NET "CE"  LOC = "p20"  ; 

NET "CF"  LOC = "p17"  ; 

NET "CG"  LOC = "p83"  ; 

NET "Cin<0>"  LOC = "p58"  ; 

NET "Cin<1>"  LOC = "p74"  ; 

NET "Cin<2>"  LOC = "p59"  ; 

NET "Cin<3>"  LOC = "p75"  ; 

NET "IndClock"  LOC = "p2"  ;

NET "Reset"  LOC = "p69"  ;

Чрез горния текст сме задали:  аргументът Аin да се задава от левите четири ключа на развойната платка (свързани чрез платката с крачета P18 – Р6), аргументът Вin – от десните четири ключа на платката (свързани чрез платката с крачета Р38 – Р24), аргументът Сin – от допълнителните ключета; резултатът се изобразява в десетичен вид върху 7 – сегментните индикаторни елементи.

-Запишете информацията във файла main_Ucf.ucf  чрез  File => Save. 
9.4. Активирайте в прозореца   Sources VHDL на развойната среда главния модул Main и отново изпълнете процесите на проектиране на устройството. 

След успешно изпълнение на процесите на проектиране, прехвърлянето на файла за конфигуриране (main.bit) от компютъра вьв FPGA –чипа  на развойната платка ще извършите по USB- интерфейс с помощта на  програмата ExPort на Digilent/Adept.
За тази цел:

10. Заредете получения от развойната среда файл за конфигуриране (Programming File) в FPGA – чипа на развойната платка.

 10.1.Свържете развойната платка към компютъра, на който е развойната среда WebPACK, с помощта на кабела за  USB – интерфейс. В момента на свързването на платката към компютъра, ключът й за източник на захранващо напрежение (означен с  позиция 2 на схемата по-долу) трябва да е в положение UEXT ! След свързването на куплунзите, превключете този ключ в положение UUSB, за да получава развойната платка захранване от компютъра чрез USB.

 10.2. Стартирайте програмата: Digilent => Adept => ExPort (или кликнете върху иконата “Shortcut to Export” на DeskTop на компютъра);

 10.3. В прозореца на ExPort първоначално стартирайте Initialize Chain, и след проверката на USB - връзката между компютъра и развойната платка, която ще извърши Export, в прозореца ще бъдат изобразени двата чипа от развойната платка, които по принцип биха могли да бъдат конфигурирани (програмирани):

10.4. Кликнете върху изображението на FPGA - чипа XC3S100E,  който сте задали в началото на декларирането на проекта да бъде използван за имплементирането на устройството и върху който трябва да заредите файла за конфигуриране. 
10.5. Намерете чрез бутона Browse в прозореца на FPGA - чипа  XC3S100E   идентификатора на файла за конфигуриране от папката на вашия проект  (би трябвало да е с идентификатор main.bit);

На горния въпрос към проектанта, зададен от развойната среда, отговорете с Yes.  

С този отговор разрешавате, след като файлът за конфигуриране бъде прехвърлен от компютъра в FPGA – чипа (с помощта на тактов сигнал на интерфейса JTAG CLK), вашето устройство върху този чип, когато започне на функционира, да се синхронизира с тактовия сигнал от вашето описание - CLK.


 10.6. Прехвърлете файла за конфигуриране в FPGA - чипа чрез бутона Program Chain.

При успешно прехвърляне на файла за конфигуриране и записването му в FPGA – чипа (тоест при успешно конфигуриране на чипа), получавате съобщение от развойната среда, дадено по-долу, а освен това зелен светодиод от платката светва (от позиция 5 от схемата на платката, дадена по-горе):

11.1. Проверете работата на проектираното устройство върху развойната платка, като задавате стойности на аргументите  A, B и C от ключовете й, както е показано на следващата снимка:

11.2. За да проверите конвейерното функциониране на проектираното аритметично устройство, намалете значително честотата на тактовите импулси, които синхронизират конвейера му (това са импулсите на вътрешния за системата от устройства сигнал InternalClk, който пък се генерира от блока  ClockDivider като сигнал ClkDiv). 

В текста на ClockDivider увеличете значително индекса на тригера от брояча, от чийто изход се получават тактовите импулси – например вместо COUNT(0) , на COUNT(28).
Най-левият светодиод на развойната платка показва моментното състояние на тактовия сигнал  InternalClk ( чрез външния сигнал IndClock) – когато диодът свети, сигналът е в ‘1’ (импулс), а когато не свети – в ‘0’ (пауза). Сменяйте  стойностите на аргументите преди нарастващите фронтове на InternalClk (преди моментие на светване на диода) – тогава те се записват във входните регистри на аритметичното устройство .
Въведете три набора от стойности на аргументите в три последователни съседни тактови периода, изчакайте два тактови периода за извършване на операциите върху първия набор от стойности на аргументите и проследете след това получаването на резултатите на изхода на изчислителното устройство в три съседни тактови периода.
PAGE  
31

_1318080391.vsd
�������� 1�

�������� 2�

�������� 3�

�������� N�

T� = N.L + tit�

L�

L�

tit�

tit�

tit�

tit�

Y(i) = A(i)*B(i)+ C(i)�

�������� ->�


_1316811859.vsd
COUNT(0)


COUNT(1)


COUNT(2)


COUNT(3)


COUNT(28)


COUNT(0)


COUNT(1)


COUNT(2)


COUNT(3)


COUNT(28)


CLK


T(0)


T(1)


T(2)


T(3)


T(28)



